Curvature using xTensor

Downloading and Installing xAct

Visit the page: http://www.xact.es
Follow installation instructions on http://www.xact.es/download.html
Linux:
1. Download the tarball xAct_V.tgz (V is the version number)
2. sudo -i ; cd /usr/share/Mathematica/Applications/; tar xvfz ~/Downloads/xAct_V.tgz

Windows:
1. Download the zip file xAct_V.zip (V is the version number)
2. unzip its contents in C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\

Read the documentation:
http://www.xact.es/documentation.html
If you want to use xTensor, you will not avoid reading the full documentation. Better earlier than later: xTensorDoc.nb
The reference notebook is useful too: xTensorRefGuide.nb

The documentation is also installed locally, most likely in:
Linux: /usr/share/Mathematica/Applications/xAct/Documentation/English/
Windows: C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\xAct\Documentation\English

Explore the documentation in xTensorDoc.nb.... Make a copy to the notebook, so that you can play with it.

```
cp /usr/share/Mathematica/Applications/xAct/Documentation/English/xTensorDoc.nb .
cp /usr/share/Mathematica/Applications/xAct/Documentation/English/xTensorRefGuide.nb .
```

Start an xTensor session

We define a 4-dim manifold M4, with metric g[-μ,-ν] = $g_{\mu\nu}$, and a Christoffel connection with covariant derivative CD[-μ][T] = $\nabla_\mu T$

In[ ]:=
```
Needs["xAct`xTensor`"]
```

```
------------------------------------------------------------
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2018, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
Package xAct`xTensor`  version 1.1.3, {2018, 2, 28}
CopyRight (C) 2002-2018, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
```

In[ ]:=
```
DefManifold[M4, 4, {λ, μ, ν, ρ, σ, α, β, γ, δ}];
DefMetric[-1, g[-μ, -ν], CD];
```

```
** DefManifold: Defining manifold M4.
** DefVBundle: Defining vbundle TangentM4.
** DefTensor: Defining symmetric metric tensor g[-μ, -ν].
** DefTensor: Defining antisymmetric tensor epsilong[-α, -β, -γ, -δ].
** DefTensor: Defining tetrametric Tetrag[-α, -β, -γ, -δ].
** DefTensor: Defining tetrametric Tetrag†[-α, -β, -γ, -δ].
** DefCovD: Defining covariant derivative CD[-μ].
** DefTensor: Defining vanishing torsion tensor TorsionCD[α, -β, -γ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[α, -β, -γ].
** DefTensor: Defining Riemann tensor RiemannCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric Ricci tensor RicciCD[-α, -β].
** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarCD[].
** DefCovD:  Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor EinsteinCD[-α, -β].
** DefTensor: Defining Weyl tensor WeylCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric TFRicci tensor TFRicciCD[-α, -β].
** DefTensor: Defining Kretschmann scalar KretschmannCD[].
** DefCovD:  Computing RiemannToWeylRules for dim 4
** DefCovD:  Computing RicciToTFRicci for dim 4
** DefCovD:  Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining weight +2 density Detg[]. Determinant.
```

The Riemann tensor

First define some toy tensors u^μ, v^μ, ξ_μ, ω_μ, $F^{\mu\nu}$, $S_{\mu\nu}$

In[ ]:=
```
DefTensor[u[ μ]    , M4]; DefTensor[v[ μ], M4]; DefTensor[w[ μ], M4];
DefTensor[ξ[-μ]    , M4]; DefTensor[ω[-μ], M4];
DefTensor[F[ μ,  ν], M4, Antisymmetric[{μ, ν}]];
DefTensor[S[-μ, -ν], M4, Symmetric[{-μ, -ν}]];
```

```
** DefTensor: Defining tensor u[μ].
** DefTensor: Defining tensor v[μ].
** DefTensor: Defining tensor w[μ].
** DefTensor: Defining tensor ξ[-μ].
** DefTensor: Defining tensor ω[-μ].
** DefTensor: Defining tensor F[μ, ν].
** DefTensor: Defining tensor S[-μ, -ν].
```

Monoterm symmetries of the Riemann tensor are built in. To enforce them in an expression you have to act with ToCanonical

In[ ]:=
```
Print[
 RiemannCD[-μ, -ν, -λ, σ] v[μ] u[ν] w[λ] ω[-σ], "\n",
 RiemannCD[-μ, -ν, -λ, σ] v[μ] v[ν] w[λ] ω[-σ], " = ",
 RiemannCD[-μ, -ν, -λ, σ] v[μ] v[ν] w[λ] ω[-σ] // ToCanonical
]
```

$R[\nabla]_{\mu\nu\lambda}{}^{\sigma}\ u^{\nu}\ v^{\mu}\ w^{\lambda}\ \omega_{\sigma}$

$R[\nabla]_{\mu\nu\lambda}{}^{\sigma}\ v^{\mu}\ v^{\nu}\ w^{\lambda}\ \omega_{\sigma} = 0$

Multiterm symmetries of the Riemann tensor are harder to implement. Read section 9.2 of xTensorDoc.nb

In[ ]:=
```
eq1 = RiemannCD[-λ, -μ, -ν, σ] + RiemannCD[-ν, -λ, -μ, σ] + RiemannCD[-μ, -ν, -λ, σ];
Print[
 eq1, "\n",
 eq1 // ToCanonical, "\n",
 eq1 // Simplification
]
```

$R[\nabla]_{\lambda\mu\nu}{}^{\sigma} + R[\nabla]_{\mu\nu\lambda}{}^{\sigma} + R[\nabla]_{\nu\lambda\mu}{}^{\sigma}$

$R[\nabla]_{\lambda\mu\nu}{}^{\sigma} - R[\nabla]_{\lambda\nu\mu}{}^{\sigma} + R[\nabla]_{\lambda}{}^{\sigma}{}_{\mu\nu}$

$R[\nabla]_{\lambda\mu\nu}{}^{\sigma} - R[\nabla]_{\lambda\nu\mu}{}^{\sigma} + R[\nabla]_{\lambda}{}^{\sigma}{}_{\mu\nu}$

Antisymmetrize can do the same work:

In[ ]:=
```
Print[
 3 Antisymmetrize[RiemannCD[-λ, -μ, -ν, σ], {-λ, -μ, -ν}], " = ",
 3 Antisymmetrize[RiemannCD[-λ, -μ, -ν, σ], {-λ, -μ, -ν}] // Simplification
]
```

$$\frac{1}{2}\left(R[\nabla]_{\lambda\mu\nu}{}^{\sigma} - R[\nabla]_{\lambda\nu\mu}{}^{\sigma} - R[\nabla]_{\mu\lambda\nu}{}^{\sigma} + R[\nabla]_{\mu\nu\lambda}{}^{\sigma} + R[\nabla]_{\nu\lambda\mu}{}^{\sigma} - R[\nabla]_{\nu\mu\lambda}{}^{\sigma}\right) = R[\nabla]_{\lambda\mu\nu}{}^{\sigma} - R[\nabla]_{\lambda\nu\mu}{}^{\sigma} + R[\nabla]_{\lambda}{}^{\sigma}{}_{\mu\nu}$$

Contravariant Riemann has more symmetries:

In[ ]:=
```
Print[
 "\nR_{μν[ρσ]} = ", 2 Antisymmetrize[RiemannCD[-μ, -ν, -ρ, -σ], {-ρ, -σ}] // ToCanonical,
 "\nR_{μν(ρσ)} = ", 2    Symmetrize[RiemannCD[-μ, -ν, -ρ, -σ], {-ρ, -σ}] // ToCanonical,
 "\nR_{[μν]ρσ} = ", 2 Antisymmetrize[RiemannCD[-μ, -ν, -ρ, -σ], {-μ, -ν}] // ToCanonical,
 "\nR_{(μν)ρσ} = ", 2    Symmetrize[RiemannCD[-μ, -ν, -ρ, -σ], {-μ, -ν}] // ToCanonical,
 "\nR_{μνρσ}-R_{ρσμν}=", RiemannCD[-μ, -ν, -ρ, -σ] - RiemannCD[-ρ, -σ, -μ, -ν] // ToCanonical
]
```

$R_{\mu\nu[\rho\sigma]} = 2\, R[\nabla]_{\mu\nu\rho\sigma}$

$R_{\mu\nu(\rho\sigma)} = 0$

$R_{[\mu\nu]\rho\sigma} = 2\, R[\nabla]_{\mu\nu\rho\sigma}$

$R_{(\mu\nu)\rho\sigma} = 0$

$R_{\mu\nu\rho\sigma}-R_{\rho\sigma\mu\nu}=0$

The last line is zero because:

$R_{\mu\lambda\sigma}\, \epsilon^{\mu\lambda\alpha\beta}\, R^{\nu\sigma} =$

$R_{\lambda\sigma\mu\nu}\, \epsilon^{\mu\lambda\alpha\beta}\, R^{\nu\sigma} = -R_{\lambda\sigma\mu\nu}\, \epsilon^{\lambda\mu\alpha\beta}\, R^{\nu\sigma} = -R_{\mu\sigma\lambda\nu}\, \epsilon^{\mu\lambda\alpha\beta}\, R^{\nu\sigma} = -R_{\mu\nu\lambda\sigma}\, \epsilon^{\mu\lambda\alpha\beta}\, R^{\sigma\nu} = -R_{\mu\nu\lambda\sigma}\, \epsilon^{\mu\lambda\alpha\beta}\, R^{\nu\sigma} = 0$

In[ ]:=
```
Print[
 RiemannCD[-μ, -ν, -λ, -σ] v[μ] u[ν] w[λ] w[σ], " = ",
 RiemannCD[-μ, -ν, -λ, -σ] v[μ] u[ν] w[λ] w[σ]  // ToCanonical, "\n",
 RiemannCD[-μ,  μ, -λ, -σ], " = ",
 RiemannCD[-μ,  μ, -λ, -σ] // ToCanonical, "\n",
 RiemannCD[-μ, -ν, -λ,  λ], " = ",
 RiemannCD[-μ, -ν, -λ,  λ] // ToCanonical, "\n",
 RiemannCD[-μ, -ν, -λ, -σ] RicciCD[μ, ν], " = ",
 RiemannCD[-μ, -ν, -λ, -σ] RicciCD[μ, ν] // ToCanonical, "\n",
 RiemannCD[-μ, -ν, -λ, -σ] EinsteinCD[λ, σ], " = ",
 RiemannCD[-μ, -ν, -λ, -σ] EinsteinCD[λ, σ] // ToCanonical, "\n",
 RiemannCD[-μ, -ν, -λ, -σ] RicciCD[ν, σ] epsilong[μ, λ, α, β], " = ",
 RiemannCD[-μ, -ν, -λ, -σ] RicciCD[ν, σ] epsilong[μ, λ, α, β] // ToCanonical
 (* you must use R_{μνλσ} = R_{λσμν}  to show that this is 0*)
]
```

$R[\nabla]_{\mu\nu\lambda\sigma}\; u^{\nu}\; v^{\mu}\; w^{\lambda}\; w^{\sigma} = 0$

$R[\nabla]_{\mu}{}^{\mu}{}_{\lambda\sigma} = 0$

$R[\nabla]_{\mu\nu\lambda}{}^{\lambda} = 0$

$R[\nabla]^{\mu\nu}\; R[\nabla]_{\mu\nu\lambda\sigma} = 0$

$G[\nabla]^{\lambda\sigma}\; R[\nabla]_{\mu\nu\lambda\sigma} = 0$

$\epsilon g^{\mu\lambda\alpha\beta}\; R[\nabla]^{\nu\sigma}\; R[\nabla]_{\mu\nu\lambda\sigma} = 0$

Contractions of the Riemann Tensor:

$R_{\mu\nu} = R_{\mu\alpha\nu}{}^{\alpha}$, $R = R_{\mu}{}^{\mu}$

In[ ]:=
```
Print[
 "Ricci tensor:      ", RicciCD[-μ, -ν], " = ", RiemannCD[-μ, -α, -ν, α], "\n",
 "Ricci scalar:      ", RicciScalarCD[], " = ", RicciCD[-μ, μ], "\n",
 "Einstein tensor:   ", EinsteinCD[-μ, -ν], " = ",
 RicciCD[-μ, -ν] - (1 / 2) g[-μ, -ν] RicciScalarCD[], " = ",
 RicciCD[-μ, -ν] - (1 / 2) g[-μ, -ν] RicciScalarCD[] // RicciToEinstein // ToCanonical,
 " = ",
 EinsteinCD[-μ, -ν] // EinsteinToRicci, "\n",
 "Trace free Ricci:  ", TFRicciCD[-μ, -ν], " = ",
 TFRicciCD[-μ, -ν] // TFRicciToRicci, ",     S_μ^μ= ", TFRicciCD[-μ, μ], "\n",
 "Weyl  tensor:      ", WeylCD[-μ, -ν, -ρ, -σ],
 " = ", WeylCD[-μ, -ν, -ρ, -σ] // WeylToRiemann
]
```

Ricci tensor: $R[\nabla]_{\mu\nu} = R[\nabla]_{\mu\nu}$

Ricci scalar: $R[\nabla] = R[\nabla]$

Einstein tensor: $G[\nabla]_{\mu\nu} = R[\nabla]_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R[\nabla] = G[\nabla]_{\mu\nu} = R[\nabla]_{\mu\nu} - \frac{1}{2} g_{\nu\mu} R[\nabla]$

Trace free Ricci: $S[\nabla]_{\mu\nu} = R[\nabla]_{\mu\nu} - \frac{1}{4} g_{\nu\mu} R[\nabla]$, $\quad S_{\mu}{}^{\mu} = 0$

Weyl tensor: $W[\nabla]_{\mu\nu\rho\sigma} = -\frac{1}{2} g_{\sigma\nu} R[\nabla]_{\mu\rho} + \frac{1}{2} g_{\rho\nu} R[\nabla]_{\mu\sigma} +$
$\frac{1}{2} g_{\sigma\mu} R[\nabla]_{\nu\rho} - \frac{1}{2} g_{\rho\mu} R[\nabla]_{\nu\sigma} - \frac{1}{6} g_{\rho\nu} g_{\sigma\mu} R[\nabla] + \frac{1}{6} g_{\rho\mu} g_{\sigma\nu} R[\nabla] + R[\nabla]_{\mu\nu\rho\sigma}$

RiemannToChristoffel

$R_{\mu\nu\lambda}{}^{\sigma} = -\partial_{\mu} \Gamma^{\sigma}{}_{\nu\lambda} + \partial_{\nu} \Gamma^{\sigma}{}_{\mu\lambda} - \Gamma^{\sigma}{}_{\mu\alpha} \Gamma^{\alpha}{}_{\nu\lambda} + \Gamma^{\sigma}{}_{\nu\alpha} \Gamma^{\alpha}{}_{\mu\lambda} = -\mathcal{R}^{\sigma}{}_{\lambda\mu\nu}$, where $\mathcal{R}^{\sigma}{}_{\lambda\mu\nu}$ is the Riemann tensor defined in Carroll+Hartle’s book

In[◦]:=
```
Print[
 RiemannCD[-μ, -ν, -λ, σ] , " = ",
 RiemannCD[-μ, -ν, -λ, σ]                    // RiemannToChristoffel  //
  ScreenDollarIndices, "\n",
 RicciCD[-μ, -ν]         , " = ",
 RicciCD[-μ, -ν]                             // RiemannToChristoffel  //
  ScreenDollarIndices, "\n",
 RicciScalarCD[]         , " = ",
 RicciScalarCD[]                             // RiemannToChristoffel  //
  ScreenDollarIndices, "\n",
 EinsteinCD[-μ, -ν]      , " = ",
 EinsteinCD[-μ, -ν]      // EinsteinToRicci  // RiemannToChristoffel  //
  ScreenDollarIndices, "\n",
 WeylCD[-μ, -ν, -ρ, -σ]  , " = ",
 WeylCD[-μ, -ν, -ρ, -σ]  // WeylToRiemann    // RiemannToChristoffel  //
  ScreenDollarIndices
]
```

$$R[\nabla]_{\mu\nu\lambda}{}^{\sigma} = -\Gamma[\nabla]^{\alpha}{}_{\nu\lambda}\,\Gamma[\nabla]^{\sigma}{}_{\mu\alpha} + \Gamma[\nabla]^{\alpha}{}_{\mu\lambda}\,\Gamma[\nabla]^{\sigma}{}_{\nu\alpha} - \partial_{\mu}\Gamma[\nabla]^{\sigma}{}_{\nu\lambda} + \partial_{\nu}\Gamma[\nabla]^{\sigma}{}_{\mu\lambda}$$

$$R[\nabla]_{\mu\nu} = -\Gamma[\nabla]^{\alpha}{}_{\mu\beta}\,\Gamma[\nabla]^{\beta}{}_{\alpha\nu} + \Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,\Gamma[\nabla]^{\beta}{}_{\mu\nu} + \partial_{\alpha}\Gamma[\nabla]^{\alpha}{}_{\mu\nu} - \partial_{\mu}\Gamma[\nabla]^{\alpha}{}_{\alpha\nu}$$

$$R[\nabla] = \left(g^{\alpha\beta}\left(\Gamma[\nabla]^{\gamma}{}_{\gamma\delta}\,\Gamma[\nabla]^{\delta}{}_{\alpha\beta} - \Gamma[\nabla]^{\gamma}{}_{\alpha\delta}\,\Gamma[\nabla]^{\delta}{}_{\gamma\beta} - \partial_{\alpha}\Gamma[\nabla]^{\gamma}{}_{\gamma\beta} + \partial_{\gamma}\Gamma[\nabla]^{\gamma}{}_{\alpha\beta}\right)\right)$$

$$G[\nabla]_{\mu\nu} = -\Gamma[\nabla]^{\alpha}{}_{\mu\beta}\,\Gamma[\nabla]^{\beta}{}_{\alpha\nu} + \Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,\Gamma[\nabla]^{\beta}{}_{\mu\nu} - \frac{1}{2}\,g_{\nu\mu}\left(g^{\alpha\beta}\left(\Gamma[\nabla]^{\gamma}{}_{\gamma\delta}\,\Gamma[\nabla]^{\delta}{}_{\alpha\beta} - \Gamma[\nabla]^{\gamma}{}_{\alpha\delta}\,\Gamma[\nabla]^{\delta}{}_{\gamma\beta} - \partial_{\alpha}\Gamma[\nabla]^{\gamma}{}_{\gamma\beta} + \partial_{\gamma}\Gamma[\nabla]^{\gamma}{}_{\alpha\beta}\right)\right) + \partial_{\alpha}\Gamma[\nabla]^{\alpha}{}_{\mu\nu} - \partial_{\mu}\Gamma[\nabla]^{\alpha}{}_{\alpha\nu}$$

$$\begin{aligned}
W[\nabla]_{\mu\nu\rho\sigma} = {} & -\frac{1}{6}\,g_{\rho\nu}\,g_{\sigma\mu}\left(g^{\alpha\beta}\left(\Gamma[\nabla]^{\gamma}{}_{\gamma\delta}\,\Gamma[\nabla]^{\delta}{}_{\alpha\beta} - \Gamma[\nabla]^{\gamma}{}_{\alpha\delta}\,\Gamma[\nabla]^{\delta}{}_{\gamma\beta} - \partial_{\alpha}\Gamma[\nabla]^{\gamma}{}_{\gamma\beta} + \partial_{\gamma}\Gamma[\nabla]^{\gamma}{}_{\alpha\beta}\right)\right) + \\
& \frac{1}{6}\,g_{\rho\mu}\,g_{\sigma\nu}\left(g^{\alpha\beta}\left(\Gamma[\nabla]^{\gamma}{}_{\gamma\delta}\,\Gamma[\nabla]^{\delta}{}_{\alpha\beta} - \Gamma[\nabla]^{\gamma}{}_{\alpha\delta}\,\Gamma[\nabla]^{\delta}{}_{\gamma\beta} - \partial_{\alpha}\Gamma[\nabla]^{\gamma}{}_{\gamma\beta} + \partial_{\gamma}\Gamma[\nabla]^{\gamma}{}_{\alpha\beta}\right)\right) - \\
& \frac{1}{2}\,g_{\sigma\nu}\left(-\Gamma[\nabla]^{\alpha}{}_{\mu\beta}\,\Gamma[\nabla]^{\beta}{}_{\alpha\rho} + \Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,\Gamma[\nabla]^{\beta}{}_{\mu\rho} + \partial_{\alpha}\Gamma[\nabla]^{\alpha}{}_{\mu\rho} - \partial_{\mu}\Gamma[\nabla]^{\alpha}{}_{\alpha\rho}\right) + \\
& \frac{1}{2}\,g_{\rho\nu}\left(-\Gamma[\nabla]^{\alpha}{}_{\mu\beta}\,\Gamma[\nabla]^{\beta}{}_{\alpha\sigma} + \Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,\Gamma[\nabla]^{\beta}{}_{\mu\sigma} + \partial_{\alpha}\Gamma[\nabla]^{\alpha}{}_{\mu\sigma} - \partial_{\mu}\Gamma[\nabla]^{\alpha}{}_{\alpha\sigma}\right) + \\
& \frac{1}{2}\,g_{\sigma\mu}\left(-\Gamma[\nabla]^{\alpha}{}_{\nu\beta}\,\Gamma[\nabla]^{\beta}{}_{\alpha\rho} + \Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,\Gamma[\nabla]^{\beta}{}_{\nu\rho} + \partial_{\alpha}\Gamma[\nabla]^{\alpha}{}_{\nu\rho} - \partial_{\nu}\Gamma[\nabla]^{\alpha}{}_{\alpha\rho}\right) - \\
& \frac{1}{2}\,g_{\rho\mu}\left(-\Gamma[\nabla]^{\alpha}{}_{\nu\beta}\,\Gamma[\nabla]^{\beta}{}_{\alpha\sigma} + \Gamma[\nabla]^{\alpha}{}_{\alpha\beta}\,\Gamma[\nabla]^{\beta}{}_{\nu\sigma} + \partial_{\alpha}\Gamma[\nabla]^{\alpha}{}_{\nu\sigma} - \partial_{\nu}\Gamma[\nabla]^{\alpha}{}_{\alpha\sigma}\right) + \\
& g_{\sigma\alpha}\left(\Gamma[\nabla]^{\alpha}{}_{\nu\beta}\,\Gamma[\nabla]^{\beta}{}_{\mu\rho} - \Gamma[\nabla]^{\alpha}{}_{\mu\beta}\,\Gamma[\nabla]^{\beta}{}_{\nu\rho} - \partial_{\mu}\Gamma[\nabla]^{\alpha}{}_{\nu\rho} + \partial_{\nu}\Gamma[\nabla]^{\alpha}{}_{\mu\rho}\right)
\end{aligned}$$

Using

In[ ]:=
```
Print[
 RiemannCD[-μ, -ν, -λ, σ] , " = ",
 RiemannCD[-μ, -ν, -λ, σ] // RiemannToChristoffel // ChristoffelToGradMetric //
   ToCanonical // ScreenDollarIndices
]
```

$$R[\nabla]_{\mu\nu\lambda}{}^{\sigma} = \frac{1}{2} g^{\sigma\alpha} \partial_\alpha\partial_\mu g_{\lambda\nu} - \frac{1}{2} g^{\sigma\alpha} \partial_\alpha\partial_\nu g_{\lambda\mu} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\alpha g_{\nu\gamma} \partial_\beta g_{\lambda\mu} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\alpha g_{\mu\gamma} \partial_\beta g_{\lambda\nu} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\beta g_{\lambda\nu} \partial_\gamma g_{\mu\alpha} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\beta g_{\lambda\mu} \partial_\gamma g_{\nu\alpha} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\alpha g_{\nu\gamma} \partial_\lambda g_{\mu\beta} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\gamma g_{\nu\alpha} \partial_\lambda g_{\mu\beta} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\alpha g_{\mu\gamma} \partial_\lambda g_{\nu\beta} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\gamma g_{\mu\alpha} \partial_\lambda g_{\nu\beta} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\beta g_{\lambda\nu} \partial_\mu g_{\alpha\gamma} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\lambda g_{\nu\beta} \partial_\mu g_{\alpha\gamma} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\alpha g_{\nu\gamma} \partial_\mu g_{\lambda\beta} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\gamma g_{\nu\alpha} \partial_\mu g_{\lambda\beta} - \frac{1}{2} g^{\sigma\alpha} \partial_\mu\partial_\lambda g_{\nu\alpha} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\beta g_{\lambda\mu} \partial_\nu g_{\alpha\gamma} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\lambda g_{\mu\beta} \partial_\nu g_{\alpha\gamma} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\mu g_{\lambda\beta} \partial_\nu g_{\alpha\gamma} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\alpha g_{\mu\gamma} \partial_\nu g_{\lambda\beta} - \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\gamma g_{\mu\alpha} \partial_\nu g_{\lambda\beta} + \frac{1}{4} g^{\beta\gamma} g^{\sigma\alpha} \partial_\mu g_{\alpha\gamma} \partial_\nu g_{\lambda\beta} + \frac{1}{2} g^{\sigma\alpha} \partial_\nu\partial_\lambda g_{\mu\alpha}$$

Commutator of Derivatives

Definition of the Riemann tensor:

$[\nabla_\mu, \nabla_\nu] v^\sigma = -R_{\mu\nu\rho}{}^{\sigma} v^\rho \quad [\nabla_\mu, \nabla_\nu] \omega_\sigma = R_{\mu\nu\sigma}{}^{\rho} \omega_\rho$

Carroll+Hartle's definition:

$[\nabla_\mu, \nabla_\nu] v^\sigma = +\mathcal{R}^\sigma{}_{\rho\mu\nu} v^\rho$, so

$\mathcal{R}^\sigma{}_{\rho\mu\nu} = -R_{\mu\nu\rho}{}^{\sigma} \Rightarrow \mathcal{R}_{\sigma\rho\mu\nu} = -R_{\mu\nu\rho\sigma} \Rightarrow \mathcal{R}_{\mu\nu\sigma\rho} = -R_{\mu\nu\rho\sigma} \Rightarrow \mathcal{R}_{\mu\nu\rho\sigma} = +R_{\mu\nu\rho\sigma} = \mathcal{R}_{\rho\sigma\mu\nu}$

Use SortCovDs and CommuteCovDs to change the order of the covariant derivatives:

In[ ]:=
```
Print[
 "-----------------------------------------------------------\n",
 "[∇μ,∇ν]v^σ  = ", CD[-μ]@CD[-ν]@v[σ]  - CD[-ν]@CD[-μ]@v[σ]   ,
 "   =  ", CD[-μ]@CD[-ν]@v[σ]  - CD[-ν]@CD[-μ]@v[σ]   // SortCovDs //
  ScreenDollarIndices, "\n",
 "[∇μ,∇ν]ωσ  = ", CD[-μ]@CD[-ν]@ω[-σ]  - CD[-ν]@CD[-μ]@ω[-σ]   ,
 "   = " , CD[-μ]@CD[-ν]@ω[-σ]  - CD[-ν]@CD[-μ]@ω[-σ]   // SortCovDs //
  ScreenDollarIndices, "\n",
 "[∇μ,∇ν]Fαβ = ", CD[-μ]@CD[-ν]@F[-α, -β] - CD[-ν]@CD[-μ]@F[-α, -β],
 " = " , CD[-μ]@CD[-ν]@F[-α, -β] - CD[-ν]@CD[-μ]@F[-α, -β] // SortCovDs //
  ScreenDollarIndices, "\n",
 "-----------------------------------------------------------\n",
        CD[-μ]@CD[-ν]@v[σ]    , " = ",
 CommuteCovDs[CD[-μ]@CD[-ν]@v[σ],   CD, {-ν, -μ}]           //
  ScreenDollarIndices, "\n",
 "-----------------------------------------------------------\n"
]
```

$[\nabla_\mu,\nabla_\nu]v^\sigma = \nabla_\mu \nabla_\nu v^\sigma - \nabla_\nu \nabla_\mu v^\sigma = R[\nabla]_{\nu\mu\alpha}{}^{\sigma} \; v^\alpha$

$[\nabla_\mu,\nabla_\nu]\omega_\sigma = \nabla_\mu \nabla_\nu \omega_\sigma - \nabla_\nu \nabla_\mu \omega_\sigma = - R[\nabla]_{\nu\mu\sigma}{}^{\alpha} \; \omega_\alpha$

$[\nabla_\mu,\nabla_\nu]F_{\alpha\beta} = \nabla_\mu \nabla_\nu F_{\alpha\beta} - \nabla_\nu \nabla_\mu F_{\alpha\beta} = - F_{\gamma\beta} \; R[\nabla]_{\nu\mu\alpha}{}^{\gamma} - F_{\alpha\gamma} \; R[\nabla]_{\nu\mu\beta}{}^{\gamma}$

$\nabla_\mu \nabla_\nu v^\sigma = R[\nabla]_{\nu\mu\alpha}{}^{\sigma} \; v^\alpha + \nabla_\nu \nabla_\mu v^\sigma$

Important property of the Einstein tensor:

In[ ]:=
```
Print[
 "∇μ G^μν = ", CD[-μ][EinsteinCD[μ, ν]]
]
```

$\nabla_\mu \; G^{\mu\nu} = 0$

Other properties:

In[ ]:=
```
Print[
 "R^μ_νμσ = ", RiemannCD[ μ, -ν, -μ, -σ], "\n",
 "W^μ_νμσ = ", WeylCD   [ μ, -ν, -μ, -σ], "\n",
 "S_μ^μ   = ", TFRicciCD[-μ, μ]
]
```

$R^{\mu}{}_{\nu\mu\sigma}$ = $R[\nabla]_{\nu\sigma}$

$W^{\mu}{}_{\nu\mu\sigma}$ = 0

$S_{\mu}{}^{\mu}$ = 0

Bianchi Identities

Taken from xTensor_Paris_A.nb, p16 (there they are also shown for covariant derivatives with torsion)

First Bianchi Identity:

In[ ]:=
```
Print[
 "6 R_[αβγ]^δ = ",
 eq1 = 6 Antisymmetrize[RiemannCD[-α, -β, -γ, δ], {-α, -β, -γ}], " = ",
 eq1 // RiemannToChristoffel // ChristoffelToGradMetric // ToCanonical
]
```

$6\ R_{[\alpha\beta\gamma]}{}^{\delta} = R[\nabla]_{\alpha\beta\gamma}{}^{\delta} - R[\nabla]_{\alpha\gamma\beta}{}^{\delta} - R[\nabla]_{\beta\alpha\gamma}{}^{\delta} + R[\nabla]_{\beta\gamma\alpha}{}^{\delta} + R[\nabla]_{\gamma\alpha\beta}{}^{\delta} - R[\nabla]_{\gamma\beta\alpha}{}^{\delta} = 0$

Second Bianchi identity:

In[ ]:=
```
Print[
 "6 ∇_[α R_βγ]δ^ν = ",
 eq1 = 6 Antisymmetrize[ CD[-α][RiemannCD[-β, -γ, -δ, ν]], {-α, -β, -γ}], " = ",
 eq1 // RiemannToChristoffel // CovDToChristoffel // ToCanonical
]
```

$6\ \nabla_{[\alpha}\ R_{\beta\gamma]\delta}{}^{\nu} = \nabla_{\alpha} R[\nabla]_{\beta\gamma\delta}{}^{\nu} - \nabla_{\alpha} R[\nabla]_{\gamma\beta\delta}{}^{\nu} - \nabla_{\beta} R[\nabla]_{\alpha\gamma\delta}{}^{\nu} + \nabla_{\beta} R[\nabla]_{\gamma\alpha\delta}{}^{\nu} + \nabla_{\gamma} R[\nabla]_{\alpha\beta\delta}{}^{\nu} - \nabla_{\gamma} R[\nabla]_{\beta\alpha\delta}{}^{\nu} = 0$

Assignment/Substitution

Functions: IndexSet, IndexSetDelayed, IndexRule, IndexRuleDelayed

Global assignment, like x = y, using IndexSet

In[ ]:=
```
(*UndefTensor[ttmp1];UndefTensor[ttmp2];UndefTensor[ttmp3];*)
```

```
In[∘]:= DefTensor[ttmp1[-μ], M4, PrintAs → "t"];
DefTensor[ttmp2[-μ, -ν, ρ], M4, PrintAs → "T"];
DefTensor[ttmp3[-μ], M4, PrintAs → "q"];
```

** DefTensor: Defining tensor ttmp1[-μ].

** DefTensor: Defining tensor ttmp2[-μ, -ν, ρ].

** DefTensor: Defining tensor ttmp3[-μ].

```
In[∘]:= IndexSet[ttmp1[-μ_], F[-μ, -ν] v[ν]];
Print[
 ttmp1[μ] // ScreenDollarIndices, "      ", ttmp1[-μ] // ScreenDollarIndices, "\n",
 ttmp1[-μ] ttmp1[-ν] g[μ, ν] // ScreenDollarIndices, " = ",
 ttmp1[-μ] ttmp1[-ν] g[μ, ν] // ContractMetric // ScreenDollarIndices, "\n",
 ttmp1[-μ] ttmp1[ μ] // ScreenDollarIndices
]
```

t^{μ} $\qquad F_{\mu\alpha}\ v^{\alpha}$

$F_{\mu\alpha}\ F_{\nu\beta}\ g^{\mu\nu}\ v^{\alpha}\ v^{\beta} = F_{\mu\alpha}\ F^{\mu}{}_{\beta}\ v^{\alpha}\ v^{\beta}$

$F_{\mu\alpha}\ t^{\mu}\ v^{\alpha}$

```
In[∘]:= IndexSet[ttmp1[-μ_], F[-μ, -ν] v[ν]]; IndexSet[ttmp1[ μ_], F[ μ, -ν] v[ν]];
Print[
 ttmp1[μ] // ScreenDollarIndices, "      ", ttmp1[-μ] // ScreenDollarIndices, "\n",
 ttmp1[-μ] ttmp1[-ν] g[μ, ν] // ScreenDollarIndices, " = ",
 ttmp1[-μ] ttmp1[-ν] g[μ, ν] // ContractMetric // ScreenDollarIndices, "\n",
 ttmp1[-μ] ttmp1[ μ] // ScreenDollarIndices
]
```

$F^{\mu}{}_{\alpha}\ v^{\alpha}$ $\qquad F_{\mu\alpha}\ v^{\alpha}$

$F_{\mu\alpha}\ F_{\nu\beta}\ g^{\mu\nu}\ v^{\alpha}\ v^{\beta} = F_{\mu\alpha}\ F^{\mu}{}_{\beta}\ v^{\alpha}\ v^{\beta}$

$F_{\mu\alpha}\ F^{\mu}{}_{\beta}\ v^{\alpha}\ v^{\beta}$

```
In[∘]:=  IndexSetDelayed[ttmp2[-μ_, -ν_, ρ_]  ,      F[-μ, -ν] ttmp3[ρ]];
Print[ttmp2[-μ, -ν, ρ] // ScreenDollarIndices];
 IndexSetDelayed[ttmp3[ μ_]           ,      S[ μ, -ν]     v[ν]];
Print[ttmp2[-μ, -ν, ρ] // ScreenDollarIndices];
 IndexSetDelayed[ttmp3[ μ_]          , RicciCD[ μ, -ν]     v[ν]];
Print[ttmp2[-μ, -ν, ρ] // ScreenDollarIndices];
```

$F_{\mu\nu}\ q^{\rho}$

$F_{\mu\nu}\ S^{\rho}{}_{\alpha}\ v^{\alpha}$

$F_{\mu\nu}\ R[\nabla]^{\rho}{}_{\alpha}\ v^{\alpha}$

Rules: make substitutions without assignments

In[∘]:=
```
{v[ ν],
 v[ ν]        /. IndexRule[ v[ μ_], g[ μ, ν] ω[-ν]] // ScreenDollarIndices,
 ω[-ν],
 ω[-ν]        /. IndexRule[ ω[-μ_], g[-μ, -ν] v[ ν]] // ScreenDollarIndices}
```

Out[∘]= $\{v^{\nu},\ g^{\nu\alpha}\ \omega_{\alpha},\ \omega_{\nu},\ g_{\nu\alpha}\ v^{\alpha}\}$

The symbol ↦ has been introduced. It is input as \[RightTeeArrow],

In[∘]:=
```
{ω[μ] /. ω[-μ_] ↦ g[-μ, -ν] v[ν],
 ω[μ] /. ω[ μ_] ↦ g[ μ, -ν] v[ν]}
```

Out[∘]= $\{\omega^{\mu},\ v^{\mu}\}$

Mapping at specific positions: use xAct/ExpressionManipulation.m

In[∘]:=
```
<< xAct/ExpressionManipulation.m
```

In[∘]:=
```
eq1 = RiemannCD[-μ, -ν, -ρ, σ] u[μ] v[ν] w[ρ] +
   F[-μ, -ν] u[μ] RicciCD[ν, σ] + WeylCD[μ, σ] ω[-μ]
```

Out[∘]= $F_{\mu\nu}\ R[\nabla]^{\nu\sigma}\ u^{\mu} + R[\nabla]_{\mu\nu\rho}{}^{\sigma}\ u^{\mu}\ v^{\nu}\ w^{\rho} + W[\nabla]^{\mu\sigma}\ \omega_{\mu}$

In[∘]:=
```
eq1 // ColorTerms
```

Out[∘]= {1} $F_{\mu\nu}\ R[\nabla]^{\nu\sigma}\ u^{\mu}$ + {2} $R[\nabla]_{\mu\nu\rho}{}^{\sigma}\ u^{\mu}\ v^{\nu}\ w^{\rho}$ + {3} $W[\nabla]^{\mu\sigma}\ \omega_{\mu}$

We can evaluate a function on given term: (xTensor_Paris_C.nb, page 5)

In[∘]:=
```
MapAt[RiemannToChristoffel, eq1, {1}] // ScreenDollarIndices
```

Out[∘]= $R[\nabla]_{\mu\nu\rho}{}^{\sigma}\ u^{\mu}\ v^{\nu}\ w^{\rho} + W[\nabla]^{\mu\sigma}\ \omega_{\mu} + F_{\mu\nu}\ g^{\nu\alpha}\ g^{\sigma\beta}\ u^{\mu}\ (\Gamma[\nabla]^{\gamma}{}_{\gamma\delta}\ \Gamma[\nabla]^{\delta}{}_{\alpha\beta} - \Gamma[\nabla]^{\gamma}{}_{\alpha\delta}\ \Gamma[\nabla]^{\delta}{}_{\gamma\beta} - \partial_{\alpha}\Gamma[\nabla]^{\gamma}{}_{\gamma\beta} + \partial_{\gamma}\Gamma[\nabla]^{\gamma}{}_{\alpha\beta})$

In[◦]:=
```
MapAt[RiemannToChristoffel, eq1, {{1}, {2}}] // ScreenDollarIndices
```

Out[◦]=
$$W[\nabla]^{\mu\sigma}\ \omega_\mu + F_{\mu\nu}\ g^{\nu\alpha}\ g^{\sigma\beta}\ u^\mu\ \left(\Gamma[\nabla]^\gamma{}_{\gamma\delta}\ \Gamma[\nabla]^\delta{}_{\alpha\beta} - \Gamma[\nabla]^\gamma{}_{\alpha\delta}\ \Gamma[\nabla]^\delta{}_{\gamma\beta} - \partial_\alpha\Gamma[\nabla]^\gamma{}_{\gamma\beta} + \partial_\gamma\Gamma[\nabla]^\gamma{}_{\alpha\beta}\right) + u^\mu\ v^\nu\ w^\rho\ \left(-\Gamma[\nabla]^\alpha{}_{\nu\rho}\ \Gamma[\nabla]^\sigma{}_{\mu\alpha} + \Gamma[\nabla]^\alpha{}_{\mu\rho}\ \Gamma[\nabla]^\sigma{}_{\nu\alpha} - \partial_\mu\Gamma[\nabla]^\sigma{}_{\nu\rho} + \partial_\nu\Gamma[\nabla]^\sigma{}_{\mu\rho}\right)$$

Find a pattern in an expression:

In[◦]:=
```
eq1 // ColorPositionsOfPattern[_RiemannCD]
```

Out[◦]=
$$F_{\mu\nu}\ R[\nabla]^{\nu\sigma}\ u^\mu + \left(\{2,1\}\ R[\nabla]_{\mu\nu\rho}{}^\sigma\right) u^\mu\ v^\nu\ w^\rho + W[\nabla]^{\mu\sigma}\ \omega_\mu$$

In[◦]:=
```
MapAt[RiemannToChristoffel, eq1, Position[eq1, _RiemannCD]] // ScreenDollarIndices
```

Out[◦]=
$$F_{\mu\nu}\ R[\nabla]^{\nu\sigma}\ u^\mu + W[\nabla]^{\mu\sigma}\ \omega_\mu + u^\mu\ v^\nu\ w^\rho\ \left(-\Gamma[\nabla]^\alpha{}_{\nu\rho}\ \Gamma[\nabla]^\sigma{}_{\mu\alpha} + \Gamma[\nabla]^\alpha{}_{\mu\rho}\ \Gamma[\nabla]^\sigma{}_{\nu\alpha} - \partial_\mu\Gamma[\nabla]^\sigma{}_{\nu\rho} + \partial_\nu\Gamma[\nabla]^\sigma{}_{\mu\rho}\right)$$

Use MakeRule to make rules. Notice that MakeRule does not use patterns

In[◦]:=
```
rule1 = MakeRule[{ω[-μ] , RicciCD[-μ, -ν] u[ν] }];
Print[
 ω[-ν] u[ν]     , " = ",
 ω[-ν] u[ν]     /. rule1                    // ScreenDollarIndices
]
```

$$u^\nu\ \omega_\nu\ =\ R[\nabla]_{\nu\alpha}\ u^\alpha\ u^\nu$$

In[ ]:=
```
rule2 = MakeRule[{ ω[-μ]      , CD[-ν][F[-μ, -ρ] u[ρ] u[ν]]},
   MetricOn → All, ContractMetrics → True, UseSymmetries → True];
rule3 = MakeRule[{ CD[-μ][u[ν]], 0}];
Print[
 ω[-μ] u[μ]                                                              ,
 " = ",
 ω[-μ] u[μ]     /. rule2                          //
  ScreenDollarIndices   , " = ",
 ω[-μ] u[μ]     /. rule2         // Simplification       //
  ScreenDollarIndices    , "    ",
 "(simplification notices that u^μu^β
   ∇_αF_μβ = 0 due to antisymmetry\n\nWe set ∇_μu^ν→0\n",
 ω[-μ] u[μ]     /. rule2 /. rule3                         //
  ScreenDollarIndices   , " = ",
 ω[-μ] u[μ]     /. rule2 /. rule3 // Simplification       // ScreenDollarIndices
]
```

$u^{\mu}\ \omega_{\mu}\ =\ u^{\mu}\left(u^{\alpha}\ u^{\beta}\left(\nabla_{\alpha} F_{\mu\beta}\right)+F_{\mu\beta}\ u^{\beta}\left(\nabla_{\alpha} u^{\alpha}\right)+F_{\mu\beta}\ u^{\alpha}\left(\nabla_{\alpha} u^{\beta}\right)\right)\ =\ F_{\beta\mu}\ u^{\alpha}\ u^{\beta}\left(\nabla_{\alpha} u^{\mu}\right)$

(simplification notices that $u^{\mu}u^{\beta}\ \nabla_{\alpha}F_{\mu\beta}\ =\ 0$ due to antisymmetry

We set $\nabla_{\mu}u^{\nu}\to 0$

$u^{\alpha}\ u^{\beta}\ u^{\mu}\left(\nabla_{\alpha} F_{\mu\beta}\right)\ =\ 0$

Geodesic Deviation

Acceleration: $v^{\rho}\nabla_{\rho}(v^{\nu}\nabla_{\nu}u^{\mu})$.

Rules:

vgeod: $v^{\nu}\nabla_{\nu}v^{\mu}\to 0$ (geodesic)

vu2uv: $v^{\nu}\nabla_{\nu}u^{\mu}\to u^{\nu}\nabla_{\nu}v^{\mu}$ ($[v, u] = 0$)

uv2vu: $u^{\nu}\nabla_{\nu}v^{\mu}\to v^{\nu}\nabla_{\nu}u^{\mu}$

rleib: $v^{\rho}\nabla_{\mu}\nabla_{\nu}v^{\sigma}\to\nabla_{\mu}(v^{\rho}\nabla_{\nu}v^{\sigma})-(\nabla_{\mu}v^{\rho})(\nabla_{\nu}v^{\sigma})$ (simple Leibniz rule)

gleib: $v^{\rho}\nabla_{\mu}\nabla_{\rho}v^{\sigma}\to\nabla_{\mu}(v^{\rho}\nabla_{\rho}v^{\sigma})-(\nabla_{\mu}v^{\rho})(\nabla_{\rho}v^{\sigma})=-(\nabla_{\mu}v^{\rho})(\nabla_{\rho}v^{\sigma})$ (Leibniz rule + contraction + $v^{\rho}\nabla_{\rho}v^{\sigma}=0$)

In[ ]:=
```
vgeod = MakeRule[{v[ρ] CD[-ρ] [v[μ]], 0                        },
    MetricOn → All, ContractMetrics → True, UseSymmetries → True];
vu2uv = MakeRule[{v[ρ] CD[-ρ] [u[μ]], u[ρ] CD[-ρ] [v[μ]]},
    MetricOn → All, ContractMetrics → True, UseSymmetries → True];
uv2vu = MakeRule[{u[ρ] CD[-ρ] [v[μ]], v[ρ] CD[-ρ] [u[μ]]},
    MetricOn → All, ContractMetrics → True, UseSymmetries → True];
rleib = MakeRule[{v[ρ] CD[-μ][CD[-ν][v[σ]]], CD[-μ][v[ρ] CD[-ν][v[σ]]] - CD[-μ][v[ρ]] CD[-ν][v[σ]]},
    MetricOn → All, ContractMetrics → True, UseSymmetries → True];
gleib =
   MakeRule[{v[ρ] CD[-μ][CD[-ρ][v[σ]]],                                      -CD[-μ][v[ρ]] CD[-ρ][v[σ]]},
    MetricOn → All, ContractMetrics → True, UseSymmetries → True];
Print[
 " v^ρ ∇_ρ(v^ν ∇_ν u^μ) =
   "                                                                                    ,
 "\n                  = ",
 tmp1 = v[ρ] CD[-ρ] @(v[ν] CD[-ν]@ u[μ] /. vu2uv) // Expand             //
    ScreenDollarIndices , "\n                      = ",
 tmp2 = tmp1 /. vu2uv                                                          //
    ScreenDollarIndices , "\n                      = ",
 tmp3 = CommuteCovDs[tmp2, CD, {-α, -ρ}]            // Expand             //
    ScreenDollarIndices  , "\n                      = ",
 tmp4 = tmp3 /. gleib                                                          //
    ScreenDollarIndices  , "\n                      = ",
 tmp4 // ToCanonical,
 "\n-----------------------------------------------------------------"
]
```

$v^{\rho}\nabla_{\rho}(v^{\nu}\nabla_{\nu}u^{\mu}) =$

$$= v^{\rho}\left(\nabla_{\alpha}v^{\mu}\right)\left(\nabla_{\rho}u^{\alpha}\right) + u^{\alpha}\ v^{\rho}\left(\nabla_{\rho}\nabla_{\alpha}v^{\mu}\right)$$

$$= u^{\beta}\left(\nabla_{\alpha}v^{\mu}\right)\left(\nabla_{\beta}v^{\alpha}\right) + u^{\alpha}\ v^{\rho}\left(\nabla_{\rho}\nabla_{\alpha}v^{\mu}\right)$$

$$= R[\nabla]_{\alpha\rho\beta}{}^{\mu}\ u^{\alpha}\ v^{\beta}\ v^{\rho} + u^{\alpha}\ v^{\rho}\left(\nabla_{\alpha}\nabla_{\rho}v^{\mu}\right) + u^{\beta}\left(\nabla_{\alpha}v^{\mu}\right)\left(\nabla_{\beta}v^{\alpha}\right)$$

$$= R[\nabla]_{\alpha\rho\beta}{}^{\mu}\ u^{\alpha}\ v^{\beta}\ v^{\rho} + u^{\beta}\left(\nabla_{\alpha}v^{\mu}\right)\left(\nabla_{\beta}v^{\alpha}\right) - u^{\alpha}\left(\nabla_{\alpha}v^{\beta}\right)\left(\nabla_{\beta}v^{\mu}\right)$$

$$= -R[\nabla]^{\mu}{}_{\beta\alpha\rho}\ u^{\alpha}\ v^{\beta}\ v^{\rho}$$

Killing Vectors

Show that $\nabla_{\mu}\nabla_{\nu}\xi_{\lambda} = R_{\lambda\nu\mu}{}^{\rho}\,\xi_{\rho}$

Difficulty: there is no algorithm at present in `xTensor`` to canonicalize expressions with multiterm symmetries, like those of the Riemann tensor. See section 9.2 in xTensorDoc.nb

```
In[ ]:= rkill = MakeRule[{CD[-μ] [ξ[-ν]] , - CD[-ν] [ξ[-μ]]  }];
Print[
 tmp0  = CD[-μ] @ CD[-ν]@ξ[-λ],
 " = "                                                        ,
 "\n       =", (**)
 tmp1 = CommuteCovDs[tmp0, CD, {-ν, -μ}] //
   ScreenDollarIndices                                    ,
 "\n       =", (*Commute Derivatives*)
 tmp2  = tmp1 /.
   rkill
         , "\n       =", (*Use Killing Equation*)
 tmp3  = CommuteCovDs[tmp2, CD, {-λ, -ν}] //
   ScreenDollarIndices                                    ,
 "\n       =", (*Commute Derivatives*)
 tmp4  = tmp3 /.
   rkill
         , "\n       =", (*Use Killing Equation*)
 tmp5  = CommuteCovDs[tmp4, CD, {-μ, -λ}] //
   ScreenDollarIndices                                    ,
 "\n       =", (*Commute Derivatives*)
 tmp6  = tmp5 /.
   rkill
         , "\n       =", (*Use Killing Equation*)
 tmp7  = tmp6 // ToCanonical

                                                                ,
 "\n       =", (*Put indices in order*)
 tmp8  = tmp7 /.  RiemannCD[-λ, -α, -μ, -ν] →
    RiemannCD[-μ, -ν, -λ, -α]                              ,
 "\n       =", (*Use symmetry Rλαμν=Rμνλα . We use simple rule
  since we want only one term substituted*)
 tmp9  = tmp8  /.  RiemannCD[-λ, -μ, -ν, -α] →
     -RiemannCD[-ν, -λ, -μ, -α] - RiemannCD[-μ, -ν, -λ, -α] // Expand,
 "\n       =", (*We use muliterm symmetry R[λμν]α=0*)
 tmp9 // ToCanonical (*Only one index needs to be permuted, ToCanonical
  knows about it. What remains is the identity that we want to prove*)
]
```

$\nabla_\mu \nabla_\nu \xi_\lambda$ =

$=- R[\nabla]_{\nu\mu\lambda}{}^{\alpha}\ \xi_\alpha + \nabla_\nu \nabla_\mu \xi_\lambda$

$=- R[\nabla]_{\nu\mu\lambda}{}^{\alpha}\ \xi_\alpha - \nabla_\nu \nabla_\lambda \xi_\mu$

$= R[\nabla]_{\lambda\nu\mu}{}^{\alpha}\ \xi_\alpha - R[\nabla]_{\nu\mu\lambda}{}^{\alpha}\ \xi_\alpha - \nabla_\lambda \nabla_\nu \xi_\mu$

$= R[\nabla]_{\lambda\nu\mu}{}^{\alpha}\ \xi_\alpha - R[\nabla]_{\nu\mu\lambda}{}^{\alpha}\ \xi_\alpha + \nabla_\lambda \nabla_\mu \xi_\nu$

$= R[\nabla]_{\lambda\nu\mu}{}^{\alpha}\ \xi_\alpha - R[\nabla]_{\mu\lambda\nu}{}^{\alpha}\ \xi_\alpha - R[\nabla]_{\nu\mu\lambda}{}^{\alpha}\ \xi_\alpha + \nabla_\mu \nabla_\lambda \xi_\nu$

$= R[\nabla]_{\lambda\nu\mu}{}^{\alpha}\ \xi_\alpha - R[\nabla]_{\mu\lambda\nu}{}^{\alpha}\ \xi_\alpha - R[\nabla]_{\nu\mu\lambda}{}^{\alpha}\ \xi_\alpha - \nabla_\mu \nabla_\nu \xi_\lambda$

$= R[\nabla]_{\lambda\alpha\mu\nu}\ \xi^\alpha + R[\nabla]_{\lambda\mu\nu\alpha}\ \xi^\alpha + R[\nabla]_{\lambda\nu\mu\alpha}\ \xi^\alpha - \nabla_\mu \nabla_\nu \xi_\lambda$

$= R[\nabla]_{\lambda\mu\nu\alpha}\ \xi^\alpha + R[\nabla]_{\lambda\nu\mu\alpha}\ \xi^\alpha + R[\nabla]_{\mu\nu\lambda\alpha}\ \xi^\alpha - \nabla_\mu \nabla_\nu \xi_\lambda$

$= R[\nabla]_{\lambda\nu\mu\alpha}\ \xi^\alpha - R[\nabla]_{\nu\lambda\mu\alpha}\ \xi^\alpha - \nabla_\mu \nabla_\nu \xi_\lambda$

$=2\ R[\nabla]_{\lambda\nu\mu\alpha}\ \xi^\alpha - \nabla_\mu \nabla_\nu \xi_\lambda$

We have shown that: $\nabla_\mu \nabla_\nu \xi_\lambda = 2 R_{\lambda\nu\mu\alpha} \xi^\alpha - \nabla_\mu \nabla_\nu \xi_\lambda \Rightarrow \quad 2 \nabla_\mu \nabla_\nu \xi_\lambda = 2 R_{\lambda\nu\mu\alpha} \xi^\alpha \Rightarrow \quad \nabla_\mu \nabla_\nu \xi_\lambda = R_{\lambda\nu\mu\alpha} \xi^\alpha$

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr

Web: http://physics.ntua.gr/konstant